2.3 Periodic Lattice

The periodic lattice problem ([F171, problem 15]) serves to demonstrate the remarkable feature of what happens to the discrete eigenvalues of bound-state problems, if the potential well is replicated many times to form an infinite lattice. This problem is relevant in solid-state physics, particularly in 3D and 2D in order to explain electron band structure and conductivity.

From a very simple mathematical toy problem in 1D we can learn how electronic delocalization comes about in a metal or even a macromolecule. The periodic arrangement of square-well potentials can be imagined to represent the potential provided by a lattice of ions to the outermost electrons. The idealization of a realistic, smoothly varying attractive potential by a discontinuous piecewise constant potential is implemented in order to allow for a simple solution by matching, as in the single square-well problem.

We consider a periodic potential in 1D that represents a sequence of square wells of width a separated by segments of length b. We set a basic potential region originating at 0 to be of the following type:

```
> Vpot:=(x,a,b,V0)->Heaviside(x+b)*Heaviside(-x);
```

$$\mathrm{Vpot} := (\,x, a, b,\ V0\,) \to \mathrm{Heaviside}(\,x + b\,)\,\mathrm{Heaviside}(\,-x\,)$$

```
> plot(Vpot(x,2,1,1),x=-2..4);
```



The interval of periodicity equals $l = a + b$. We plot three adjacent segments of the potential.

```
> plot(Vpot(x,2,1,1)+Vpot(x+3,2,1,1)+Vpot(x-3,2,1,1),x=-5..5);
```



Naively one might think that this is a replica of the standard square-well problem. But there are some surprises. First, one uses an argument to relate the neighbouring problems, by requesting that the probability distribution be the same for the electron within each cell. Thus, their wavefunctions can differ by at most a phase factor

$$\psi(x+l) = \beta\psi(x) \rightarrow \psi(x+nl) = \beta^n\psi(x)\,.$$

The phase factor can be expressed in terms of the lattice period l and a constant to cancel the dimension: $\beta = \exp(\mathrm{i}Kl)$, with K dimensionally a wave number. The periodicity in the function $\exp(\mathrm{i}\phi)$ allows us to restrict the range of the propagation number (which is a vector if considered in 3D) to the base interval

$$-\pi < Kl < \pi\,,$$

thereby introducing a reduced propagation number.

With the assumption that $E < V0$, $k^2 = 2mE/\hbar^2$, $q^2 = 2m(V0-E)/\hbar^2$ we write down the wavefunctions for the first and second cells (in units in which $m = \hbar = 1$):

```
> V0:=1;    a:=2;    b:=1;
```

$$V0 := 1 \quad a := 2 \quad b := 1$$

```
> ks:=sqrt(2*En);      qs:=sqrt(2*(V0-En));
```

$$ks := \sqrt{2}\sqrt{En} \quad qs := \sqrt{2-2\,En}$$

```
> assume(En>0);
```

ψ_{11} is the inside-the-barrier part, ψ_{21} is the free solution – both are for cell 1.

```
> psi11:=x->A1*exp(qs*x)+B1*exp(-qs*x);
```

$$\psi 11 := x \to A1\,e^{(qs\,x)} + B1\,e^{(-qs\,x)}$$

```
> psi21:=x->A2*exp(I*ks*x)+B2*exp(-I*ks*x);
```

$$\psi 21 := x \to A2\,e^{(I\,ks\,x)} + B2\,e^{(-I\,ks\,x)}$$

Now we define the same states in the neighbouring cell: lp is the periodicity length, Kp the propagation number. Note how the argument in parts of ψ_{12} and ψ_{22} is shifted!

```
> lp:=a+b;
```

$$lp := 3$$

```
> psi12:=x->exp(I*Kp*lp)*(A1*exp(qs*(x-lp))+B1*exp(-qs*(x-lp)));
```

$$\psi 12 := x \to e^{(I\,Kp\,lp)}\,(A1\,e^{(qs\,(x-lp))} + B1\,e^{(-qs\,(x-lp))})$$

```
> psi22:=x->exp(I*Kp*lp)*(A2*exp(I*ks*(x-lp))
> +B2*exp(-I*ks*(x-lp)));
```

$$\psi 22 := x \to e^{(I\,Kp\,lp)}\,(A2\,e^{(I\,ks\,(x-lp))} + B2\,e^{(-I\,ks\,(x-lp))})$$

Now we set up the matching conditions. The four constants Ai, Bi are determined from 4 matching conditions, which, however, represent a homogeneous system of equations.

```
> eq1:=psi11(0)=psi21(0);
```

$$\text{eq1} := A1 + B1 = A2 + B2$$

```
> eq2:=D(psi11)(0)=D(psi21)(0);
```

$$\text{eq2} := A1\,\sqrt{2-2\,En^{\sim}} - B1\,\sqrt{2-2\,En^{\sim}} = I\,A2\,\sqrt{2}\,\sqrt{En^{\sim}} - I\,B2\,\sqrt{2}\,\sqrt{En^{\sim}}$$

Now we match onto the next cell at $x = a$:

```
> eq3:=psi21(a)=psi12(a);
```

$$\text{eq3} := A2\,e^{(2\,I\,\sqrt{2}\,\sqrt{En^{\sim}})} + B2\,e^{(-2\,I\,\sqrt{2}\,\sqrt{En^{\sim}})} = e^{(3\,I\,Kp)}\left(A1\,e^{(-\sqrt{2-2\,En^{\sim}})} + B1\,e^{(\sqrt{2-2\,En^{\sim}})}\right)$$

```
> eq4:=D(psi21)(a)=D(psi12)(a);
```

$$eq4 := I\,A2\,\sqrt{2}\,\sqrt{En^{\sim}}\,e^{\left(2\,I\,\sqrt{2}\,\sqrt{En^{\sim}}\right)} - I\,B2\,\sqrt{2}\,\sqrt{En^{\sim}}\,e^{\left(-2\,I\,\sqrt{2}\,\sqrt{En^{\sim}}\right)} = e^{(3\,I\,Kp)} \left(A1\,\sqrt{2-2\,En^{\sim}}\,e^{\left(-\sqrt{2-2\,En^{\sim}}\right)} - B1\,\sqrt{2-2\,En^{\sim}}\,e^{\left(\sqrt{2-2\,En^{\sim}}\right)}\right)$$

Now we try something naive by attempting a direct solution:

```
> solve({eq1,eq2,eq3,eq4},{A1,A2,B1,B2});
```

$$\{B1 = 0, B2 = 0, A2 = 0, A1 = 0\}$$

Is this just an uninteresting trivial solution? No, we have to use the energy En to make the determinant of the coefficient matrix vanish. Whether we can find energies will depend on the choice of the propagation number Kp. In fact, the continuous variable Kp allows us to find a continuum of solutions En, i.e., the problem is qualitatively different from a single well with the potential high at the boundaries that has a discrete spectrum. It is also different from a potential that goes to zero asymptotically, which admits continuous scattering solutions at all energies $En > 0$.

How do we extract the determinant for the system of equations? We order the unknowns into $(A1, B1, A2, B2)$

```
> collect(eq2,A1);
```

$$A1\,\sqrt{2-2\,En^{\sim}} - B1\,\sqrt{2-2\,En^{\sim}} = I\,A2\,\sqrt{2}\,\sqrt{En^{\sim}} - I\,B2\,\sqrt{2}\,\sqrt{En^{\sim}}$$

```
> coeff(",A1);
Error, unable to compute coeff
```

It appears that `collect/coeff` doesn't work on equations, one can use it only on expressions. So let us redefine the equations by arranging all terms on the left: row 1 in the coefficient matrix is trivial: $1, 1, -1, -1$.

```
> eq2:=D(psi11)(0)-D(psi21)(0);
```

$$eq2 := A1\,\sqrt{2-2\,En^{\sim}} - B1\,\sqrt{2-2\,En^{\sim}} - I\,A2\,\sqrt{2}\,\sqrt{En^{\sim}} + I\,B2\,\sqrt{2}\,\sqrt{En^{\sim}}$$

```
> eq3:=expand(psi21(a)-psi12(a));
```

$$eq3 := A2\,\left(e^{\left(I\,\sqrt{2}\,\sqrt{En^{\sim}}\right)}\right)^2 + \frac{B2}{\left(e^{\left(I\,\sqrt{2}\,\sqrt{En^{\sim}}\right)}\right)^2} - \frac{\left(e^{(I\,Kp)}\right)^3 A1}{e^{\left(\sqrt{2-2\,En^{\sim}}\right)}} - \left(e^{(I\,Kp)}\right)^3 B1\,e^{\left(\sqrt{2-2\,En^{\sim}}\right)}$$

```
> eq4:=expand(D(psi21)(a)-D(psi12)(a));
```

$$eq4 := I\, A2\, \sqrt{2}\, \sqrt{En\tilde{}}\, \left(e^{\left(I\sqrt{2}\sqrt{En\tilde{}}\right)}\right)^2 - \frac{I\, B2\, \sqrt{2}\, \sqrt{En\tilde{}}}{\left(e^{\left(I\sqrt{2}\sqrt{En\tilde{}}\right)}\right)^2}$$
$$- \frac{(e^{(I\, Kp)})^3\, A1\, \sqrt{2 - 2\, En\tilde{}}}{e^{\left(\sqrt{2-2\, En\tilde{}}\right)}}$$
$$+ (e^{(I\, Kp)})^3\, B1\, \sqrt{2 - 2\, En\tilde{}}\, e^{\left(\sqrt{2-2\, En\tilde{}}\right)}$$

```
> coeff(eq2,A2);
```

$$-I\, \sqrt{2}\, \sqrt{En\tilde{}}$$

This seems to work.

```
> with(linalg):
> row1:=vector([1,1,-1,-1]);
```

$$row1 := [1\ 1\ -1\ -1]$$

```
> row2:=vector([coeff(eq2,A1),coeff(eq2,B1),coeff(eq2,A2),
> coeff(eq2,B2)]);
```

$$row2 := \left[\sqrt{2 - 2\, En\tilde{}}\quad -\sqrt{2 - 2\, En\tilde{}}\quad -I\, \sqrt{2}\, \sqrt{En\tilde{}}\quad I\, \sqrt{2}\, \sqrt{En\tilde{}}\right]$$

```
> row3:=vector([coeff(eq3,A1),coeff(eq3,B1),coeff(eq3,A2),
> coeff(eq3,B2)]);
```

$$row3 := \left[-\frac{(e^{(I\, Kp)})^3}{e^{\left(\sqrt{2-2\, En\tilde{}}\right)}}\quad -(e^{(I\, Kp)})^3\, e^{\left(\sqrt{2-2\, En\tilde{}}\right)}\quad \left(e^{\left(I\sqrt{2}\sqrt{En\tilde{}}\right)}\right)^2\right.$$
$$\left.\frac{1}{\left(e^{\left(I\sqrt{2}\sqrt{En\tilde{}}\right)}\right)^2}\right]$$

```
> row4:=vector([coeff(eq4,A1),coeff(eq4,B1),coeff(eq4,A2),
> coeff(eq4,B2)]);
```

$$row4 := \left[-\frac{(e^{(I\, Kp)})^3\, \sqrt{2 - 2\, En\tilde{}}}{e^{\left(\sqrt{2-2\, En\tilde{}}\right)}}\right.$$
$$(e^{(I\, Kp)})^3\, \sqrt{2 - 2\, En\tilde{}}\, e^{\left(\sqrt{2-2\, En\tilde{}}\right)}$$

$$I\sqrt{2}\sqrt{En\tilde{}}\left(e^{(I\sqrt{2}\sqrt{En\tilde{}})}\right)^2 - \frac{I\sqrt{2}\sqrt{En\tilde{}}}{\left(e^{(I\sqrt{2}\sqrt{En\tilde{}})}\right)^2}\Bigg]$$

```
> Coeff:=matrix(4,4,0):
> for i from 1 to 4 do
> Coeff[1,i]:=row1[i];    Coeff[2,i]:=row2[i];
> Coeff[3,i]:=row3[i];    Coeff[4,i]:=row4[i];
> od:
> evalm(Coeff);
```

$$[1, 1, -1, -1]$$
$$\left[\sqrt{2-2\,En\tilde{}}, -\sqrt{2-2\,En\tilde{}}, -I\sqrt{2}\sqrt{En\tilde{}}, I\sqrt{2}\sqrt{En\tilde{}}\right]$$
$$\left[-\frac{(e^{(I\,Kp)})^3}{\%1}, -(e^{(I\,Kp)})^3\,\%1, \%2^2, \frac{1}{\%2^2}\right]$$
$$\left[-\frac{(e^{(I\,Kp)})^3\sqrt{2-2\,En\tilde{}}}{\%1}, (e^{(I\,Kp)})^3\sqrt{2-2\,En\tilde{}}\,\%1,\right.$$
$$\left.I\sqrt{2}\sqrt{En\tilde{}}\,\%2^2, -\frac{I\sqrt{2}\sqrt{En\tilde{}}}{\%2^2}\right]$$
$$\%1 := e^{(\sqrt{2-2\,En\tilde{}})}$$
$$\%2 := e^{(I\sqrt{2}\sqrt{En\tilde{}})}$$

```
> chareq:=det(Coeff);
```

$$\text{chareq} := 2\Big(2\,I\,\%2\sqrt{2-2\,En\tilde{}}\sqrt{2}\sqrt{En\tilde{}}\,\%1^2$$
$$-2\,(e^{(I\,Kp)})^3\,\%2^2\,En\tilde{} + 2\,(e^{(I\,Kp)})^3\,\%2^2\,En\tilde{}\,\%1^4$$
$$-I\,(e^{(I\,Kp)})^3\sqrt{2-2\,En\tilde{}}\,\%2^2\sqrt{2}\sqrt{En\tilde{}}$$
$$-I\,(e^{(I\,Kp)})^3\sqrt{2-2\,En\tilde{}}\,\%2^2\sqrt{2}\sqrt{En\tilde{}}\,\%1^4$$
$$+(e^{(I\,Kp)})^3\,\%2^2 - \%2^2\,(e^{(I\,Kp)})^3\,\%1^4$$
$$+2\,(e^{(I\,Kp)})^3\,En\tilde{} - 2\,(e^{(I\,Kp)})^3\,En\tilde{}\,\%1^4$$
$$-I\,(e^{(I\,Kp)})^3\sqrt{2-2\,En\tilde{}}\sqrt{2}\sqrt{En\tilde{}}$$
$$-I\,(e^{(I\,Kp)})^3\sqrt{2-2\,En\tilde{}}\sqrt{2}\sqrt{En\tilde{}}\,\%1^4$$
$$+2\,I\,(e^{(I\,Kp)})^6\sqrt{2-2\,En\tilde{}}\,\%2\sqrt{2}\sqrt{En\tilde{}}\,\%1^2$$
$$-(e^{(I\,Kp)})^3 + (e^{(I\,Kp)})^3\,\%1^4\Big)\Big/(\%2\,\%1^2)$$
$$\%1 := e^{(I\sqrt{2}\sqrt{En\tilde{}})}$$

$$\%2 := e^{\left(\sqrt{2-2\,\overline{En}}\right)}$$

```
> sol:=solve(chareq,En);
```

sol :=

That didn't work. Suspected problem: Kp isn't specified (the next fruitless attempt takes quite a while to complete).

```
> sol:=solve(subs(Kp=0,chareq),En);
```

sol :=

Now we should morally be allowed to use a numeric solver:

```
> sol:=fsolve(subs(Kp=0,chareq),En=0..V0);
```

sol := .5000000000

Apart from the occasionally occurring undesired numerical noise of an imaginary part, we have a numerical answer. However, we aren't too happy about the large value of En, as we are mostly interested in En < V0.

```
> sol:=Re(fsolve(subs(Kp=0,chareq),En=0.51..V0));
```

sol := .7550000000

Now we are ready to explore and set up a loop to obtain a list of values for plotting purposes. The numerical output of pairs (Kp_i, E_i) is removed below for $i > 5$. We explain below why we cannot solve for the roots of the magnitude of the complex-valued characteristic equation.

```
> for i from 1 to 20 do
> Kpi[i]:=evalf(-Pi+6.28/21*i)/lp;
> Ensol[i]:=fsolve(subs(Kp=Kpi[i],Re(chareq)+Im(chareq)),
> En=0.01..V0-0.01);
> print(Kpi[i],Ensol[i]);      od:
```

−.9475150116, .5849117411

−.8478324719, .5456879542

−.7481499323, .4959266352

−.6484673926, .4447491723

−.5487848526, .3970435050